


Determination of the Worst Peak Output Voltage in a Linear Time Invariant Circuit

based on the Step Response



Hypothesis:

- Input is bounded $L_{\min} \leq In \leq L_{\max}$
- Module is causal, **LINEAR** and time-invariant

```

#*
#*          s0          sn0
#* x --> (+)-----+----> [*n0/d0]----> (+)--> y
#* |           |   |
#* |           | Delay|
#* |           |   |
#* | sd1       | s1      sn1   |
#* (+)<----[*-d1/d0]<----+----> [*n1/d0]----> (+)
#* |           |   |
#* |           | Delay|
#* |           |   |
#* | sd2       | s2      sn2   |
#* (+)-----[*-d2/d0]<----+----> [*n2/d0]----> (+)
#* |           |   |
#* |           | Delay|
#* |           |   |
#* | sd3       | s3      sn3   |
#* +----[*-d3/d0]<----+----> [*n3/d0]----+
#* feedback path          feedforward path

```

We will apply the method
to an example:

the Output of this High Pass IIR Analog Filter

$$L_{min} = 0.0$$

$$L_{max} = 1.0$$

Local Extrema of the IIR Transfer Function

NAPA netlist

```
header <napatool.hdr>

title "transfer function of IIR filter"

fs      48.0e3

ivar npts POWEROF2(20)

directive REPEAT 25

node in noise 0.0 1.0
node out generator hpf3 <iir> "f3.dat" in

tool tf "tf.out" in 1.0 out 1.0 5.0e3 npts           // transfer function

terminate TOOL_INDEX >= 1
```

file "f3.dat"

```
## IIR filter, High Pass.
# numerator denominator
0 -0.986961 1.000000
1 2.960880 -2.973750
2 -2.960880 2.947840
3 0.986961 -0.974089
```

```

cell interface      $y  $dummy  $x
declare (analog)  $x


# feedforward path coefficients
dvar $n0 -9.86961000000000e-001
dvar $n1 2.96088000000000e+000
dvar $n2 -2.96088000000000e+000
dvar $n3 9.86961000000000e-001
# feedback path coefficients
dvar $d1 -2.97375000000000e+000
dvar $d2 2.94784000000000e+000
dvar $d3 -9.74089000000000e-001
# signal path
node $s1 delay    $s0
node $s2 delay    $s1
node $s3 delay    $s2

node $sd1 gain   -$d1  $s1
node $sd2 gain   -$d2  $s2
node $sd3 gain   -$d3  $s3
node $sn0 gain   $n0  $s0
node $sn1 gain   $n1  $s1
node $sn2 gain   $n2  $s2
node $sn3 gain   $n3  $s3

node $s0 sum     $sd1  $sd2  $sd3  $x
node $y sum     $sn0  $sn1  $sn2  $sn3


```

IIR Filter Transfer Function



We will use this IIR filter to show the method of determination of the worst case of peak.

1. Apply a step function



2. Use the step response to compute the worst case of peak using the theorem of superposition for linear systems

IIR Filter Step Response

NAPA netlist


```
header <napatool.hdr>

fs          48.0e3

title "Step Response of IIR Filter"

node in   step 0.0 1.0 1.0e-6
node out  generator hpf3 <iir> "f3.dat" in
node stb  iuser stable out 10    1.0e-6      // check for signal stabilization
output "resp.out" in  out           // step response
terminate (LOOP_INDEX >= 5000) || stb
```

IIR Filter Step Response



Local Extrema of the IIR Filter Step Response

NAPA netlist

```
header <napatool.hdr>

fs          48.0e3

title "Local Extrema of the IIR Filter Step Response"


node in   step  0.0  1.0  1.0e-6
node out  generator hpf3 <iir>    "f3.dat"  in

node stb  iuser stable out 10    1.0e-6           // check for signal stabilization

output "resp.out"  in  out          // step response
post  extrema "ext.out" 0  2  (short)        // list of extrema

terminate  (LOOP_INDEX >= 5000) || stb
```

Local Extrema of the IIR Filter Step Response




Method A

- *Indirect Method* -

Use Time Position of Local Extrema
of the IIR Filter Step Response
to build activations leading to
the Worst Case Peak voltage.


Method A: Use Time Position of Local Extrema of the IIR Filter Step Response



Build an activation in such a way that all maximums occur at the same time.

Superposition of maximums

Method A: Use Time Position of Local Extrema of the IIR Filter Step Response



Method A: Use Time Position of Local Extrema of the IIR Filter Step Response

NAPA netlist

```
header <napatool.hdr>

fs          48.0e3

title "Build activation for Worst Case of Peak of IIR filter"


debug WCPK_PWL

node in   step 0.0 1.0 1.0e-6
node out generator hpf3 <iir>    "f3.dat"  in

node stb  iuser stable out 10  1.0e-6    // check for signal stabilization

output "resp.out"  in  out           // step response
post  wcpk_pwl "wcpk_pwl.out" 0  2    // compute the 2 activations to be used
                                         // to get this worst case of peak
terminate (LOOP_INDEX >= 5000) || stb
```

Method A: Use Time Position of Local Extrema of the IIR Filter Step Response



Use Computed Activation to get the Worst Case of Peak

NAPA netlist

```
header <napatool.hdr>

title "Use computed PWL to get the Worst Case of Peak"

fs          48.0e3                                // same sampling frequency !!!

debug PWL


node  inp  duser pwl  "wcpk_pwl.out"    TIME 1  (aperiodic)
node  inm  duser pwl  "wcpk_pwl.out"    TIME 2  (aperiodic)
node  outp generator hpf3p  <iir>      "f3.dat"  inp
node  outm generator hpf3m  <iir>      "f3.dat"  inm

node  stbp iuser stable outp  100  1.0e-4
node  stbm iuser stable outm  100  1.0e-4


output "worst.out"  inp inm  outp outm

terminate  (LOOP_INDEX > 10000) || (stbm && stbp)
```


Minimum Peak Voltage of IIR Filter



Maximum Peak Voltage of IIR Filter



Minimum and Maximum Peak Voltage of IIR Filter




Method B


- *Direct Method* -

Use Local Extremum Voltages
of the IIR Filter Step Response
to get **directly** the value of the
Worst Case Peaks.

Method B: Use Local Extremum Voltages of the IIR Filter Step Response



Method B: Use Local Extremum Voltages of the IIR Filter Step Response



Method B: Use Local Extremum Voltages of the IIR Filter Step Response

NAPA netlist

```
header <napatool.hdr>

fs          48.0e3

title "Direct Worst Case of Peak evaluation of IIR filter"

debug WCPK

node in   step 0.0 1.0 1.0e-6
node out  generator hpf3 <iir>    "f3.dat"  in

node stb  iuser stable out 10    1.0e-6      // check for signal stabilization

output "resp.out"  in  out           // step response
post   wcpk        "wcpk.out"  0  2      // compute the worst case of peak

terminate  (LOOP_INDEX >= 5000) || stb
```

file "wcpk.out"

```
# F3_D
# (tool           ) wcpk postprocessor applied on "resp.out"
# (compiler version) NAPA V2.81
# (source file     ) f3_d.mac
# (random seed     ) 585223739
# (postprocessed file) resp.out
# (postprocessed column) #1, #3
# (postprocessed tool) output
#
# (postprocessed signal) current_time -> out
#
# [ Compute the WORST CASE OF PEAK from a Step Response ]
#
# Fri Feb 02 21:17:59 2007 by YVES LEDUC
#      min_peak          max_peak          peak_to_peak
-1.270794e+000    1.270766e+000    2.541560e+000
# end of output file
```